Data Structures – List Boxes and Lists
A list box control displays a list of string items in a box. It provides a number of facilities, including the ability to select an item in a list by clicking on it, add items, and delete items. The list box control is available along with the other controls on the toolbox and can be placed as usual on a form. We will use as an example a shopping list, building it up by adding items one by one. After some items have been added, the list box looks like Figure 1.  Each item occupies a single line. If the complete list cannot be displayed in the available space, a scroll bar is automatically displayed. Later we will see how to delete items from the list.

List boxes are a good introduction to using data structures because they provide a direct, visual representation of the information. 

Figure 1. A List box

Lists

When we use a list box by placing it on the form from the toolbox, we are creating a new instance of the ListBox class. The ListBox class makes use of another class, called a List, to carry out its functions. A list box merely displays information on the form and handles mouse click events, but a list actually holds the information displayed in a text box. So while a list box supports the events Click and DoubleClick and properties such as SelectedItem, a list provides methods to add and remove items from the list.

For example, if we create a list box named Shopping, then the property Shopping.Items is the list containing the information displayed in the list box:

Dim myList As List

myList = Shopping.Items

We can then use the properties and methods of lists with myList. For example, we can obtain a count of the number of items in the list (and in the list box) as follows:

Dim numberOfItems As Integer

numberOfItems = myList.Count

This series of statements can be written more concisely as follows:

Dim numberOfItems As Integer

numberOfItems = Shopping.Items.Count

in which we have chosen not to explicitly mention the name of the list.

Adding items to a list

The example program shown in Figure 2 allows the user to add items to a list box.

Figure 2. Adding items to a shopping list

The following method responds to a button click and places an item of shopping at the end of the list box.

Private Sub Button1_Click(ByVal sender As System.Object, _

                          ByVal e As System.EventArgs) _

                          Handles Button1.Click

    Shopping.Items.Add(TextBox1.Text)

End Sub

In this example the name of the list box is Shopping. As we have seen, one of the properties of a list box is Items and this property represents the contents of the list box as an instance of the List class. This class in turn provides a number of methods, one of which is the Add method that allows items to be added to a list. Its parameter is the value to be added to the list. It must be a string.

Another way of placing items in a list box is to do it at design time. Selecting the Items property of a list box throws up a new window in which items can be inserted into the list box. 

The length of a list

Next, here is a method that responds to a button click and displays a message box containing the number of items currently in the list box.

Private Sub CountButton_Click(ByVal sender As System.Object, _

                              ByVal e As System.EventArgs) _

                              Handles CountButton.Click

    MessageBox.Show(CStr(Shopping.Items.Count))

End Sub

Again we see how the property Items of the list box named Shopping is used. In turn the property Count of the List class is used to obtain the number of items held in the list box.

Indices

A program refers to the items in a list box by an index. An index is an integer that says which item is being referred to. The first item has index value 0, the second 1, etc. We can visualize the above list box as a table as shown in Figure 3, with the index values alongside (but not actually stored with the data).

	0
	bread

	1
	milk

	2
	coffee


Figure 3. Diagram of a list box showing the indexes.

We now look at a program that emphasizes and demonstrates index values. The user clicks on an item in a list box and the program displays the equivalent index value in a text box (Figure 4). 

Figure 4 Selecting an item from a list box

When the click event arrives, the following method is called to handle the event. 

Private Sub Shopping_MouseClick( _

                   ByVal sender As System.Object, _

                   ByVal e As System.EventArgs) _

                   Handles Shopping.SelectedIndexChanged

    TextBox1.Text = CStr(Shopping.SelectedIndex)

End Sub

SelectedIndex is a list box property that provides the index value of the item clicked on (or -1 if nothing has been selected). Running this program emphasizes that the index values are not actually stored as part of a list box, but that the computer knows the values and they can be used as and when necessary. You also confirm, when you run this program, that the index values start at zero (not at 1).

Figure 5 shows a program that allows the user to display the item corresponding to a chosen index value. The code to handle the button click is:

Private Sub Button1_Click(ByVal sender As System.Object, _

                          ByVal e As System.EventArgs) _

                          Handles Button1.Click

    Dim index As Integer

    index = CInt(IndexTextBox.Text)

    ValueTextBox.Text = CStr(Shopping.Items(index))
End Sub

The program extracts the value from the list box using the expression:
Shopping.Items(index)
In this expression, Shopping is the name of the list box. Items is the property of a list box that gives the contents of the list box, which is a list. Finally the index value is placed in brackets after the name of the list. Thus, for example

Shopping.Items(2)
would give us the value in the list box at index value 2.

Figure 5 Displaying an item from a list box

Removing items from a list

We have seen how to add items to a list box. Now we consider removing information. The method RemoveAt of the class List removes the item at a particular index value. So if we have a list box Shopping, we can remove the item at index value 3 by:

Shopping.Items.RemoveAt(3)

When this happens, the gap created is closed up.

Inserting items within a list

We have seen how to add items to the end of a list using the method Add. It is also possible to insert items within the body of a list, using method Insert. Given an existing list, we can for example do this:

Shopping.Items.Insert(5, "tea")

The item formerly at index value 5 is moved down the list, along with any subsequent items.

Week 9

Lookup

A table such a list box is conveniently used for lookup. For example, we can construct a list box (Figure 6) that contains the names of the months, January to December. Then if someone gives us a month expressed as a number (1 to 12) we can use the table to convert the number to the equivalent text.

	0
	January

	1
	February

	2
	March

	3
	etc


Figure 6. Diagram of a list box for converting integers to month names.

Figure 7. The month conversion program

Figure 7 shows how the program looks to its user. We will make the list box invisible (by setting its Visible property to False), since there is no need for the user of the program to know about it.

When the program is designed, we enter the values January, February, March, etc. directly into the Items property of the list box.

When the program runs, a number entered via a text box can be converted as follows:

Private Sub Button1_Click(ByVal sender As System.Object, _

                          ByVal e As System.EventArgs) _

                          Handles Button1.Click

    Dim monthNumber As Integer

    Dim monthName As String

    monthNumber = CInt(MonthNumberTextBox.Text)

    monthName = CStr(Months.Items(monthNumber - 1))
    MonthNameTextBox.Text = monthName

End Sub
The numbers representing a month run from 1 to 12, whereas index values start at 0. Therefore we need to subtract 1 from the month number, as shown, to convert it into an appropriate index. The Items property of a list box allows the program to access the value of an item in the list box named Months.

Using a lookup table as above is an alternative to writing a series of If statements to carry out the conversion, which has the following structure:

If monthNumber = 1 Then

    monthName = "January"

Else

    If monthNumber = 2 Then

        monthName = "February"

    End If

End If
Yet another alternative would be to use a Select Case statement. Employing If statements or a Select Case statement makes use of actions to carry out the conversion. In contrast, using a table (such as a list box) embodies the conversion information more neatly within the table.

Arithmetic on a list box

We now look at a list box, named Numbers, that contains integer numbers and we will carry out arithmetic on the numbers. A list box always contains strings, but one kind of string is a string of digits – a number. Figure 8 shows a program that allows its user to enter numbers into a list box. Then one button causes the sum of the numbers to be displayed and another button causes the largest number to be displayed.

Figure 8 Arithmetic on a list box

Here is the program to add together all the numbers in a list. A For statement is used to run through all the values of the index. Remember index values start at 0. The index of the last item in the list is equal to the length of the list - 1. Each value in the list is added to a running total, called Sum, which is initially made equal to 0. Finally the value is placed in a text box.

Private Sub SumButton_Click(ByVal sender As System.Object, _

                            ByVal e As System.EventArgs) _

                            Handles SumButton.Click

    Dim number As Integer

    Dim index As Integer

    Dim sum As Integer

    sum = 0

    For index = 0 To Numbers.Items.Count - 1

        number = CInt(Numbers.Items(index))

        sum = sum + number

    Next 

    SumTextBox.Text = CStr(sum)

End Sub

Next we study a method to find the largest item in a list of numbers. A variable called largest is used to keep track of the largest value. Initially, it is made equal to the value at index 0 in the list box. A For statement is used to process all of the numbers in the list. Each item in the list is compared with largest, and if it is larger, the value of largest is updated.

Private Sub LargestButton_Click(ByVal sender As System.Object, _

                                ByVal e As System.EventArgs) _

                                Handles LargestButton.Click

    Dim number As Integer

    Dim index As Integer

    Dim largest As Integer

    largest = CInt(Numbers.Items(0))

    For index = 1 To Numbers.Items.Count - 1

        number = CInt(Numbers.Items(index))

        If number > largest Then

            largest = number

        End If

    Next 

    LargestTextBox.Text = CStr(largest)

End Sub

These two sections of program illustrate a common feature of programs that manipulate lists: whenever you need to process every item in a list, a For statement is the appropriate tool. Clearly a loop is needed to examine repetitively each item in what might be a long list. The alternative structures for describing a loop are the For statement and the While statement. The For statement is preferable in this case because we know at the outset of the loop how many repetitions are necessary.

Searching

This next program carries out a search. It assumes that a list (for example, the shopping list) is already set up and that we want to search the list for some item. The user enters the desired item (for example sugar) into a text box as shown in Figure 9.

Figure 9 Searching a list box

The program starts from the first item in the list and continues down the list one item at a time, trying to find the desired item. If it is not found, the index value becomes equal to the size of the list, length, and the loop ends. If the item is found, the Boolean variable found is set to true and the loop terminates. A While statement is used rather than a For statement for controlling the loop, since we do not know in advance how many repetitions will be necessary.

Private Sub Button1_Click(ByVal sender As System.Object, _

                          ByVal e As System.EventArgs) _

                          Handles Button1.Click

    Dim length As Integer

    Dim index As Integer

    Dim found As Boolean

    Dim itemWanted As String

    length = Shopping.Items.Count

    itemWanted = TextBox1.Text

    found = False

index = 0

    While (found = False) And (index < length)

        If CStr(Shopping.Items(index)) = itemWanted Then

            found = True

            MessageBox.Show("Item found")

        Else

            index = index + 1

        End If

    End While

End Sub
This is a classical serial search method.

Summary

· a list box is a GUI box that contains a list of strings

· a program can add items to the end of a ListBox using the method AddItem. The ListBox automatically expands to accommodate the additional data.

· a program can remove an item from anywhere within a TextBox. The gap automatically closes.

· an item within a ListBox can be changed

· each item in a ListBox is uniquely identified by an integer, called an index. Index values are not stored. Index values always start at 0.

Exercises

1. Write a program in which an item selected in a ListBox (by clicking on it) is immediately deleted. Alternatively provide a "delete" button to delete the item that is currently selected.

2. Alter the program so that items in the ListBox are automatically always sorted into alphabetical order.

3. Add a button that causes the ListBox to be emptied, using the method Clear. 

4. Alter the program so that an item in the ListBox can be replaced by some other text. For example, "milk" is replaced by "sugar".  Provide a button marked "replace" that carries out this action. The new text is to be entered into a text area. Also provide a delete button.

5. Improve the search method so that it displays a message whether or not the required item is found in the ListBox.

